Asünkroonsed süsteemid

- Asünkroonne andmevahetus
  - andmevahetus moodulite vahel, mis kasutavad erinevaid taktsignaale
    - signaal peab olema stabiilne takti muutumisel (setup & hold)
  - andmevahetus süsteemis,
    kus signaali levimise aeg on võrreldav taktsignaali perioodiga
    - hajus-süsteemid
    - süsteem kiibil (SoC)
    - kõikjal sama probleem – signaali stabiilsuse tagamine
  - andmevahetus asünkroonsete (self-timed) moodulite vahel
    - asünkroonsed automaadid

- Asünkroonseid süsteeme tuleks kasutada seal, kus on neist kasu!

© Peeter Ellervee

Asünkroonsed sisendid

- **Signaalid moodulite vahel, mille taktsignaalid võivad oluliselt erineda**
  - **erinevate sagedustega taktsignaalid**
  - **üksteisest kaugel asuvad moodulid**
  - **füüsikalised efektid – nt. värelemine**
- **Vajalik on sisendite täiendav puhverdamine**
  - **vähendab metastabiilsuse levimise tõenäosust**









Sünkroonne või asünkroonne nullimine?

- **Sünkroonne nullimine**
  - kõikidel mäluelementidel samal ajal
    - taktsignaali aktiivne front viiakse kõikide mäluelementideni (enamvähem) samal ajal
  - takti olemasolu on vajalik
    - taktsignaal ei pruugi alguses olemas olla





- **Asünkroonne nullimine**
  - takti olemasolu ei ole vajalik
    - mõjub mäluelemendile kindlasti
  - nullimis-signaal võib eri mäluelementidel lõppeda erineval ajal
    - signaali levimine võib olla erinev – vaadeldakse tavalise signaalina, kus sama-aegsus pole kriitiline
- **Kombineeritud lähenemine – algus asünkroonne, lõpp sünkroonne**

© Peeter Ellervee

Asünkroonsed automaadid

- Taktsignaal puudub
- Muutus sisendil on sündmuseks → isetakteeruv
- Signaalide võistlus (signal race) – kriitilise tee analüüs väga oluline
  - sündmus saabub eri teid pidi – võib näida kahe või enama sündmusena
- Asünkroonne andmevahetus
- Isesünkroniseerivad süsteemid





Asünkroonsed automaadid

- Automaadi tabel (GSA, olekudiagramm) aluseks
- Olekute kodeerimine
  - nn. naaberkoodid – erinevus ühes järgus
  - vähendab signaalide võistluse ohtu siirdel





© Peeter Ellervee

Digitaalsüsteemide realiseerimine

- **Süsteemi disain – System design**
  a.k.a. **Arhitektuuri süntees – Architectural-level synthesis**
  - **Süsteemi taseme süntees**
    - **tükeldamine & liideste süntees**
  - **Kõrgtaseme süntees**
    - **planeerimine, eraldamine & sidumine**
- **Loogikadisain – Logic Design**
  - **Registersiirete taseme süntees**
    - **andme- & juhtosa süntees**
  - **Loogikataseme süntees**
    - **loogika minimeerimine & optimeerimine**
- **Füüsiline disain – Physical design**
  a.k.a. **Geomeetria süntees – Geometrical-level synthesis**
  - **Füüsikalise taseme süntees**
    - **teisendus loogikaelementideks**
    - **paigaldamine & trasseerimine/ruutimine**







© Peeter Ellervee

Tarkvara või riistvara / Üldotstarbeline või eridisain ?



- **Tarkvara ~~ üldotstarbeline disain**
  - **universaalne andmetee – aeglane, suured mõõtmed, energianäljane**
  - **universaalne algoritm – äärmiselt paindlik, kuid nõuab palju ruumi (bitte)**
- **Riistvara ~~ eridisain**
  - **fikseeritud andmetee – kiire, kompaktne, väike energiatarve**
  - **paindlikkus määratud juhtautomaadiga, algoritm sageli fikseeritud**

© Peeter Ellervee

Realisatsiooni optimaalsus?

- **operatsiooni hind riistvaras**
  - **nihutamine == traatide teisiti viimine**
  - **liitmist ja lahutamist teostab sama seade – summaator (+ülekanne)**
- **sõltumatud operatsioonid võib täita korraga**
  - **tegelike andmesõltuvuste analüüs – nt. R2 nihutamise tulemus**



Andmeosa → juhtosa

- **Osa operatsioonidest andmeosas teisendatakse operatsioonideks juhtosas**
  - **võrdlused konstantidega**
  - **siirdetingimuste avaldised (*if-then-else*, *case*, tsüklid)**
  - **optimeeritakse koos automaadiga – võib toimuda olekute arvu plahvatuslik kasv**

```
case a is
when 2 => ...
when 3 => ...
when 4 => ...
when others => ...
end case;
```





a&state:
010:000
011:000
100:000
.....

Juhtosa → andmeosa

- **Osa operatsioonidest juhtosas teisendatakse operatsioonideks andmeosas**
  - **siirdetingimuste arvutuste grupeerimine**
  - **lihtsustab automaadi sünteesi**
  - **parem (osade) loogikafunktsioonide minimeerimine**

```
if x="010" then y:="0110";
elsif x="011" and
      z="110" then y:="0101";
else
  if z="010" then y:="1001";
  else            y:="1100";
  end if;
end if;
```







Aritmeetika-operatsioonide realiseerimine

- **Summator e. liitja**
  - **poolsummator – $(c_i,o_i) = a_i + b_i$**
    - $o_i = a_i \oplus b_i$ $c_i = a_i \,\&\, b_i$
  - **täis-summator – $(c_i,o_i) = a_i + b_i + c_{i-1}$**
    - $o_i = a_i \oplus b_i \oplus c_i$ $c_i = (a_i \,\&\, b_i) \mid (a_i \,\&\, c_{i-1}) \mid (b_i \,\&\, c_{i-1})$
- **Lahutaja**
  - **täiendkood – $O = A - B = A + \overline{B} + 1$**
  - **liitja-lahutaja – $O = n \; ? \; (A - B) : (A + B)$**









© Peeter Ellervee

Aritmeetika-operatsioonide realiseerimine

- **Ülekande kiirendamine**
  - jagatud gruppideks, grupi sees ülekanne eraldi funktsioonina
  - **"carry-lookahead"**
    - "generate" – $g_i = a_i \,\&\, b_i$ "propagate" – $p_i = a_i \mid b_i$ "carry" – $c_i = g_i \mid (p_i \,\&\, c_{i-1})$
    - $g_{i+1} = a_{i+1} \,\&\, b_{i+1}$; $p_{i+1} = a_{i+1} \mid b_{i+1}$; $c_{i+1} = g_{i+1} \mid (p_{i+1} \,\&\, c_i)$
    - $c_{i+1} = g_{i+1} \mid (p_{i+1} \,\&\, (g_i \mid (p_i \,\&\, c_{i-1}))) \ldots$
  - **"carry-select"**
    - spekulatiivne arvutus – vanemates järkudes leitakse summa ja ülekanne edasi nii 0 kui 1 jaoks, tulemus valitakse multipleksoriga, kui ülekanne alt "lõpuks kohale jõuab"





Aritmeetika-operatsioonide realiseerimine

- **Korrutamine**
  - $A = a_{n-1}*2^{n-1}+a_{n-2}*2^{n-2}+...+a_2*2^2+a_1*2^1+a_0*2^0$
  - $B = b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0$
  - $A * B = a_{n-1}*2^{n-1}*(b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0)+$
    $a_{n-2}*2^{n-2}*(b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0)+...$
    $a_2*2^2*(b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0)+$
    $a_1*2^1*(b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0)+$
    $a_0*2^0*(b_{n-1}*2^{n-1}+b_{n-2}*2^{n-2}+...+b_2*2^2+b_1*2^1+b_0*2^0) =$
    $a_{n-1}*b_{n-1}*2^{2*n-2}+a_{n-1}*b_{n-2}*2^{2*n-3}+...+a_0*b_2*2^2+a_0*b_1*2^1+a_0*b_0*2^0$



Aritmeetika-operatsioonide realiseerimine

- **Korrutamine**
  - **paralleelne – kombinatsioonskeem**
    - **m*n bitti – m n-bitist summaatorit ja m*n JA-elementi**
    - **kombinatoorskeem liiga suure viitega – abiregistrid & konveier (pipeline)**
  - **järjestikuline – järk-järgu kaupa tsükliliselt**
    - **akumulaator ja juhtautomaat**
    - **mitu järku korraga – juhtloogika lisaks**
- **Jagamine – tavaliselt tsükliliselt – akumulaator, juhtloogika ja juhtautomaat**







http://www.ecs.umass.edu/ece/koren/arith/simulator/



Algoritmi realiseerimine

- Algoritm kui programm
  - operatsioonid – andmeosa
  - järjestus ja tingimuse – juhtosa
- GCD (Greatest Common Divisor) – suurim ühistegur

```
while ( x != y ) {
  if ( x < y ) y = y - x;
  else         x = x - y;
}
```

  - operatsioonid – võrdlused ja lahutamine
  - konkreetse operatsiooni realiseerimine?
    - A < B == A - B < 0 / A != B == A - B != 0 – kõike saab teha lahutajaga?!
  - sama riistvara kõikidele operatsioonidele või korraga arvutamine?
    - kiirus või suurus? [ja kas see ongi probleem?]

© Peeter Ellervee

Näide #1 – GCD (Greatest Common Divisor)

- **Spetsifikatsioon ~~ käitumuslik kirjeldus**
  - **sisendi/väljundi ajastus fikseeritud – juht- ja takt-signaalid**

```
process -- gcd-bhv.vhdl
  variable x, y: unsigned(15 downto 0);
begin
  -- Wait for the new input data
  wait on clk until clk='1' and rst='0';
  x := xi;    y := yi;    rdy <= '0';
  wait on clk until clk='1';
  -- Calculate
  while  x /= y  loop
    if  x < y  then y := y - x;
    else            x := x - y;    end if;
  end loop;
  -- Ready
  xo <= x;   rdy <= '1';
  wait on clk until clk='1';
end process;
```

Probleemid
- tsüklis puudub takt
- keerukas “wait” käsk
(- mitu “wait” käsku)

Mida proovida?
- erinevad süntesaatorid
- suuruse minimeerimine
- kiiruse tõstmine

Tehnoloogiad – ASIC, FPGA

VHDL kood & testpingid
`http://mini.pld.ttu.ee/~lrv/gcd/`

© Peeter Ellervee

GCD – sünteesitav kood?

- **Takteeritud käitumuslik stiil**

```
process -- gcd-bhvc.vhdl
  variable x, y: unsigned(15 downto 0);
begin
  -- Wait for the new input data
  while  rst = '1'  loop
    wait on clk until clk='1';
  end loop;
  x := xi;    y := yi;    rdy <= '0';
  wait on clk until clk='1';
  -- Calculate
  while  x /= y  loop
    if  x < y  then y := y - x;
    else            x := x - y;    end if;
    wait on clk until clk='1';
  end loop;
  -- Ready
  xo <= x;   rdy <= '1';
  wait on clk until clk='1';
end process;
```

ASIC: sünteesitav
961 e.g. / 20.0 ns
2 lahutajat, 2 võrdlejat

FPGA: mitte-sünteesitav
“wait” käsud tsüklis :(
juhtautomaat vajalik :(:(

Võimalikud valikud
- funktsioonide jagamine
- universaalsed funktsioonid
- spekulatiivne arvutamine

© Peeter Ellervee

GCD – käitumuslik automaat (behavioral FSM)

```
process begin -- gcd-bfsm.vhdl
  wait on clk until clk='1';
  case  state  is
  -- Wait for the new input data
  when S_wait =>
   if  rst='0'  then
     x<=xi; y<=yi; rdy<='0'; state<=S_start;
   end if;
  -- Calculate
  when S_start =>
    if  x /= y  then
      if  x < y  then  y <= y - x;
      else             x <= x - y;    end if;
      state<=S_start;
    else
      xo<=x; rdy<='1'; state<=S_ready;
    end if;
  -- Ready
  when S_ready =>    state<=S_wait;
  end case;
end process;
```

ASIC: sünteesitav
911 e.g. / 19.4 ns
2 lahutajat, 2 võrdlejat

FPGA: sünteesitav
108 SLC / 9.9 ns
2 lahutajat, 2 võrdlejat

Kas saab teha paremaks?

Jälle need võimalikud valikud
- **funktsioonide jagamine**
  - **üks operatsioon taktis**
- **universaalsed funktsioonid**
  - **A<B == A-B<0 / A/=B == A-B/=0**
- **spekulatiivne arvutamine**
  - **lahutamine enne otsustamist**

© Peeter Ellervee

GCD – universaalsed funktsioonid?

- A<B == A-B<0 / A/=B == A-B/=0

```
-- Three operations:
--   subtraction, and
--   comparisons not-equal &
--   less-than
xo <= xi - yi;

ne <= '1' when xi /= yi else '0';

lt <= '1' when xi < yi else '0';
```

ASIC: 209 e.g. / 19.9 ns
FPGA: 20 SLC / 12.1 ns
kolm liitjat!



```
-- ALU - subtracting and then comparing
x_out <= xi - yi;    xo <= x_out;
process (x_out)
  variable or_tmp: unsigned(15 downto 0);
begin
  or_tmp(15) := x_out(15);
  for  i in 14 downto 0  loop
    or_tmp(i) := or_tmp(i+1) or x_out(i);
  end loop;
  ne <= to_bit(or_tmp(0));
end process;
lt <= to_bit(x_out(15));
```

ASIC: 148 e.g. / 21.8 ns / FPGA: 12 SLC / 14.6 ns



GCD – disainiruumi uurimine

- **Erinevad lahendused – `http://mini.pld.ttu.ee/~lrv/gcd/`**
  - **gcd-bhv.vhdl – puhas käitumuslik kirjeldus, mitte-sünteesitav**
  - **gcd-bhvc.vhdl – takteeritud käitumuslik kirjeldus, osad süntesaatorid OK**
  - **gcd-bfsm.vhdl – nn. käitumuslik automaat (automaat & käitumuslik andmetee), sünteesitav (aga kui efektiivne see on?)**
  - **gcd-rtl1.vhdl – üks ALU, 3 takti iteratsiooni kohta – 1) “pole võrdne?”, 2) “väiksem kui?”, 3) lahutaja**
  - **gcd-rtl2.vhdl – üks ALU, 2 takti iteratsiooni kohta – 1) “pole võrdne?” ja “väiksem kui?”, 2) lahutaja**
  - **gcd-rtl3.vhdl – võrdleja (väiksem kui) kontrollib lahutajat, 1 takt iteratsiooni kohta – väike kuid aeglane (jadamisi) andmetee**
  - **gcd-rtl4.vhdl – spekuleeriv arvutamine – mõlemad lahutamised tehakse kõigepealt, siis toimub otsustamine (üks lahutaja võrdleb “väiksem kui”, teine “pole võrdne”)**
  - **gcd-rtl5.vhdl – spekuleeriv arvutamine – mõlemad lahutamised tehakse kõigepealt, siis toimub otsustamine (üks lahutaja võrdleb “väiksem kui”, kuid eraldi “pole võrdne”)**

© Peeter Ellervee

GCD – üksik ALU, 2 takti iteratsiooni kohta

gcd-rtl2.vhdl

```
-- Next state function of the FSM
process (state, rst, alu_ne, alu_lt) begin
  ena_x <= '0';     ena_y <= '0';       ena_r <= '0';
  set_rdy <= '0';   xi_yi_sel <= '0';   sub_y_x <= '0';
  next_state <= state;
  case state  is
  when S_wait =>    -- Wait for the new input data
    if  rst='0'  then
      xi_yi_sel <= '1';   ena_x <= '1';   ena_y <= '1';
      next_state <= S_start;
    end if;
  when S_start =>    -- Loop: ready?
    if  alu_ne='1'  then
      if  alu_lt='1'  then    next_state <= S_sub_y_x;
      else    next_state <= S_sub_x_y;    end if;
    else    next_state <= S_ready;
    end if;
  when S_sub_y_x =>    -- Loop: y-x
    ena_y <= '1';    sub_y_x <= '1';
    next_state <= S_start;
  when S_sub_x_y =>    -- Loop: x-y
    ena_x <= '1';    sub_y_x <= '0';
    next_state <= S_start;
  when S_ready =>    -- Ready
    ena_r <= '1';    set_rdy <= '1';
    next_state <= S_wait;
  end case;
end process;
```

```
-- ALU: subtract / less-than / not-equal
alu_o <= alu_1 - alu_2;
alu_lt <= to_bit(alu_o(15));
process (alu_o)
  variable or_tmp: unsigned(15 downto 0);
begin
  or_tmp(15) := alu_o(15);
  for  i in 14 downto 0  loop
    or_tmp(i) := or_tmp(i+1) or alu_o(i);
  end loop;
  alu_ne <= to_bit(or_tmp(0));
end process;

-- Multiplexers
x_i <= xi when xi_yi_sel='1' else alu_o;
y_i <= yi when xi_yi_sel='1' else alu_o;
alu_1 <= y when sub_y_x='1' else x;
alu_2 <= x when sub_y_x='1' else y;

-- Registers
process begin
  wait on clk until clk='1';
  state <= next_state;
  if  ena_x='1'  then    x <= x_i;    end if;
  if  ena_y='1'  then    y <= y_i;    end if;
  if  ena_r='1'  then    xo <= x;     end if;
  rdy <= set_rdy;
end process;
```

© Peeter Ellervee

GCD – komparaator+lahutaja, 1 takt

gcd-rtl3.vhdl

```
-- Next state function of the FSM
process (state, rst, alu_ne) begin
  ena_xy <= '0';    ena_r <= '0';
  set_rdy <= '0';    xi_yi_sel <= '0';
  next_state <= state;
  case state is
  when S_wait =>    -- Wait for the new input data
    if  rst='0'  then
      xi_yi_sel <= '1';    ena_xy <= '1';
      next_state <= S_start;
    end if;
  when S_start =>    -- Calculate
    if  alu_ne='1'  then
      ena_xy <= '1';    next_state <= S_start;
    else
      ena_r <= '1';    set_rdy <= '1';
      next_state <= S_ready;
    end if;
  when S_ready =>    -- Ready
    ena_r <= '1';    set_rdy <= '1';
    next_state <= S_wait;
  end case;
end process;

-- Comparator (less-than)
sub_y_x <= '1' when x < y else '0';
```

```
-- Subtracter (+not-equal)
alu_o <= alu_1 - alu_2;
process (alu_o)
  variable or_tmp: unsigned(15 downto 0);
begin
  or_tmp(15) := alu_o(15);
  for  i in 14 downto 0  loop
    or_tmp(i) := or_tmp(i+1) or alu_o(i);
  end loop;
  alu_ne <= to_bit(or_tmp(0));
end process;
-- Multiplexers
x_i <= xi when xi_yi_sel='1' else alu_o;
y_i <= yi when xi_yi_sel='1' else alu_o;
alu_1 <= y when sub_y_x='1' else x;
alu_2 <= x when sub_y_x='1' else y;
ena_x <= '1' when (sub_y_x='0' and ena_xy='1') or
                  xi_yi_sel='1' else '0';
ena_y <= '1' when (sub_y_x='1' and ena_xy='1') or
                  xi_yi_sel='1' else '0';
-- Registers
process begin
  wait on clk until clk='1';
  state <= next_state;
  if  ena_x='1'  then    x <= x_i;    end if;
  if  ena_y='1'  then    y <= y_i;    end if;
  if  ena_r='1'  then    xo <= x;    end if;
  rdy <= set_rdy;
end process;
```

© Peeter Ellervee

GCD – spekuleeriv arvutamine (2 lahutajat)

gcd-rtl5.vhdl

```
-- Next state function of the FSM
process (state, rst, alu_ne) begin
  ena_xy <= '0';     ena_r <= '0';
  set_rdy <= '0';    xi_yi_sel <= '0';
  next_state <= state;
  case state is
  -- Wait for the new input data
  when S_wait =>
    if  rst='0'  then
      xi_yi_sel <= '1';    ena_xy <= '1';
      next_state <= S_start;
    end if;
  -- Calculate
  when S_start =>
    if  alu_ne='1'  then
      ena_xy <= '1';
      next_state <= S_start;
    else
      ena_r <= '1';     set_rdy <= '1';
      next_state <= S_ready;
    end if;
  -- Ready
  when S_ready =>
    ena_r <= '1';     set_rdy <= '1';
    next_state <= S_wait;
  end case;
end process;
```

```
-- Subtracter (x-y) / comparator (x<y)
alu_o1 <= x - y;
sub_y_x <= '1' when alu_o1(alu_o1'high)='1' else '0';

-- Subtracter (y-x)
alu_o2 <= y - x;

-- Comparator (y/=x)
alu_ne <= '1' when x /= y else '0';

-- Multiplexers
x_i <= xi when xi_yi_sel='1' else alu_o1;
y_i <= yi when xi_yi_sel='1' else alu_o2;
ena_x <= '1' when (sub_y_x='0' and ena_xy='1') or
         xi_yi_sel='1' else '0';
ena_y <= '1' when (sub_y_x='1' and ena_xy='1') or
         xi_yi_sel='1' else '0';

-- Registers
process begin
  wait on clk until clk='1';
  state <= next_state;
  if  ena_x='1'  then    x <= x_i;    end if;
  if  ena_y='1'  then    y <= y_i;    end if;
  if  ena_r='1'  then    xo <= x;    end if;
  rdy <= set_rdy;
end process;
```

© Peeter Ellervee

GCD – sünteeside tulemused

Tehnoloogia		FPGA				ASIC			
Piirangud[1]		50 MHz		100 MHz		50 MHz		25 MHz	
	clk / FU	[SLC]	[ns]	[SLC]	[ns]	[e.g.]	[ns]	[e.g.]	[ns]
gcd-bhv[2]	*? / ?*	*93*	*17.3*	-	-	*1141*	*20.0*	-	-
gcd-bhvc	*1 / 2+2*	-	-	-	-	961	20.0	977	31.1
gcd-bfsm	1 / 2+2	108	9.9	108	9.4	911	19.4	984	30.8
gcd-rtl1	3 / 1	50	10.8	50	9.7	986	19.8	883	32.4
gcd-rtl2	2 / 1	48	10.8	48	10.0	931	19.9	882	32.3
gcd-rtl3	1 / 1+1	58	17.0	58	14.6	1134	20.0	928	40.0
gcd-rtl4	1 / 2	78	12.6	78	9.0	976	19.9	928	29.0
gcd-rtl5	1 / 2+1	58	8.0	58	7.6	915	20.0	932	26.9

[1] Taktiperiood oli ainuke piirang

[2] gcd-bhv sünteesiti prototüüp kõrgtasemesüntesaatoriga xTractor

© Peeter Ellervee

GCD – kust tulevad need erinevused?



gcd-bfsm

2 –, 1 =, 1 <, 3 rg, 3 st

1 takt/iteratsioon

108 SLC / 9.9 ns

GCD – kust tulevad need erinevused?



gcd-rtl1

1 ALU, 3 rg, 6 st

3 takti/iteratsioon

50 SLC / 10.8 ns

gcd-rtl2

1 ALU, 3 rg, 5 st

2 takti/iteratsioon

48 SLC / 10.8 ns



gcd-rtl3

1 –, 1 <, 3rg, 3st

1 takt/iteratsioon

58 SLC / 17.0 ns

© Peeter Ellervee

GCD – kust tulevad need erinevused?



gcd-rtl4

2 –, 3 rg, 3 st

1 takt/iteratsioon

78 SLC / 12.6 ns

GCD – kust tulevad need erinevused?



gcd-rtl5

2 –, 1 /=, 3 rg, 3 st

1 takt/iteratsioon

58 SLC / 8.0 ns

© Peeter Ellervee

Realisatsiooni näide #2 – täisarvude korrutamine

- Märgita täisarvude korrutamine, 2-bitti korraga (radix-4)
- o = a * b
  - … + [0,1,2,3]*b –> 3*b = 4*b - b
  - samm, kui eelmine bitipaar ei olnud ‘11’
    - b_{10}=00 ? –> ei midagi
    - b_{10}=01 ? –> o+=a
    - b_{10}=10 ? –> o+=2*a [o+=(a<<1)]
    - b_{10}=11 ? –> o-=a [ja jätame meelde]
  - samm, kui eelmine bitipaar oli ‘11’
    - b_{10}=00 ? –> o+=a [4-1==3]
    - b_{10}=01 ? –> o+=2*a [2==1+1]
    - b_{10}=10 ? –> o-=a [3==2+1, jälle meelde]
    - b_{10}=11 ? –> ei midagi [4==3+1, jälle meelde]

Korrutamine 2 kohaga

00 - blok.
01 - 1*Rg1
10 - 2*Rg1 (L1(Rg1))
11 - - Rg1; +1 järgmisesse järku

$11_2 = 100 - 1 = 10\bar{1}$
N: 0,01101101
⇩
0,10010101



y1: Rg3 := Rg3 + Rg1
y2: Rg3 := Rg3 + L1(Rg1)
y3: Rg3 := Rg3 - Rg1

Margus Kruus: IAY0140 “Arvutite aritmeetika ja loogika”

Realisatsiooni näide (ver. 1)





JA Moore: i 5, o 12, s 12, t 28; Mealy: i 5, o 12, s 4, t 13 (tagasiside OK)
OA 2 sum., 1 võrdl., 5 (6) mux, 5 (6) reg.
3 takti iteratsiooni kohta, kokku 24 (+1) takti

© Peeter Ellervee

Realisatsiooni näide (ver. 2)





JA Moore: i 5, o 12, s 10, t 82; Mealy: i 5, o 12, s 3, t 17

OA 2 sum., 1 võrdl., 5 mux, 5 reg.

1 takt iteratsiooni kohta, kokku 9 (+1) takti

© Peeter Ellervee

Realisatsiooni näide (ver. 3)

- **Keerukus: JA –> OA**
- **siirded ~~ kombinatsioonskeem**



JA Moore: i 2, o 8, s 3, t 5; Mealy: i 2, o 8, s 3, t 5
OA 2 sum., 1 võrdl., 5 mux, 5 reg., 1 (2) f-n
1 takt iteratsiooni kohta, kokku 9 (+1) takti

© Peeter Ellervee

Realisatsiooni näide (ver. 3)

- **Keerukus: OA –> JA**
- **loendur ~~ automaat**
- **Veelgi kiirem:**
  - **ühitada nullimine (y7,y5) ja esimene iteratsioon**
  - **OA: peaaegu sama**
  - **JA: 9 olekut**
  - **kiirus: 8 (+1) takti**



JA Moore: i 1, o 6, s 10, t 11; Mealy: i 1, o 6, s 10, t 11
OA 1 sum., 4 mux, 4 reg., 1 (2) f-n
1 takt iteratsiooni kohta, kokku 9 (+1) takti

© Peeter Ellervee

Realisatsiooni näide (ver. 4)

- **Eelmised variandid**
  - **2 noorimat järku kasutamat**
  - **korrektse tulemuse asukoht?**
- **Nihutamine <u>pärast</u> liitmist/lahutamist**
  - **2 vanimat järku alati 0-d!!!**
- **Pisi-erinevused OA-s**



JA Moore: i 1, o 6, s 10, t 11; Mealy: i 1, o 6, s 10, t 11
OA 1 sum., 4 mux, 4 reg., 1 (2) f-n
1 takt iteratsiooni kohta, kokku 9 (+1) takti

© Peeter Ellervee

Realisatsioonide võrdlused

- **Seitse erinevat korrutaja realisatsiooni**
  - **paralleelne, bitt korraga, 2 bitti korraga (versioonide 3 ja 4 erinevad VHDL koodid)**
  - **VHDL koodide erinevused põhjustatud sünteesi juhtimise vajadusest**
    - **“what you write is what you get...”**
  - **VHDL failid – http://mini.pld.ttu.ee/~lrv/IAY0150/multiplier/**

Tüüp	nr.	Synopsys DC		Xilinx ISE	
		comb. & reg. [e.g.]	[ns]	[slices]	[ns]
paralleelne	0	685 + 112 = 797	20.0	0 (1/24 mult.)	~10
radix-2 (bhv-rtl)	1	227 + 508 = 735	15.5	46 (7680)	6.7
radix-4 (v.4, bhv-rtl)	2	402 + 511 = 913	20.0	52 (7680)	7.8
radix-4 (v.4, ~90% rtl)	3	303 + 511 = 814	15.9	37 (7680)	7.1
radix-4 (v.4, 100% rtl)	4	179 + 511 = 690	19.9	34 (7680)	7.3
radix-4 (v.3, bhv-rtl)	5	397 + 514 = 911	20.0	51 (7680)	8.2
radix-4 (v.3, 100% rtl)	7	182 + 511 = 693	18.7	36 (7680)	6.4

© Peeter Ellervee

Seos füüsilise maailmaga

- **Sisendid, väljundid jms.**
  - vt. nt. John F. Wakerly, “Digital design: principles and practices.” Pearson/Prentice Hall

Sisendid

1) lülitid

põhiprobleemiks kontaktide *värelemine*



võimalikud lahendused







... ja loomulikult ka RC-ahel (madalpääsfilter)

Schmitti triger

2) ADC – analoog-digitaal muundurid

pinge -> kood
vool -> kood
sagedus -> kood
jne.

põhi-elemendiks komparaator võrreldakse erinevust sisend- ja tugipinge vahel

paralleelne – n bitti -> 2^n komparaatorit

järjestikuline ~~ FSM, DAC + komparaator



© Peeter Ellervee

Seos füüsilise maailmaga

- **Sisendid, väljundid jms.**

Väljundid

1) indikaatorid (nt. LED)



2) täiturid (nt. relee)



3) DAC – digitaal-analoog muundurid

nt. XSA-3S1000
www.xess.com



Siinid

sobitus



kaitsedioodid



lühisekaitse



© Peeter Ellervee